

Kapitel 26 LST-Modelle Mplus Code 2020

Mplus-Code zu Kelava, A., Schermelleh-Engel, K. & Mayer, A. (2020). Latent-State-Trait-Theorie (LST-Theorie). In H. Moosbrugger & A. Kelava (Hrsg.), Testtheorie und Fragebogenkonstruktion (3., vollständig überarbeitete und ergänzte Auflage).

Die folgende Kovarianzmatrix verwenden:

Abspeichern als: IN_Cov.txt

3.03

1.97 2.89

2.29 1.99 3.10

1.89 2.04 2.26 2.84

2.18 1.99 2.47 2.12 3.04

2.06 2.08 2.19 2.28 2.37 2.94

Model 1: Multistate-Modell

```
TITLE:      Modell 1: Multistate-Modell

DATA:      FILE = IN_Cov.txt;
           TYPE = COVARIANCE;
           NOBSERVATIONS = 302;

VARIABLE:  NAMES = Y11 Y21 Y12 Y22 Y13 Y23;

           USEVARIABLES =
             Y11 Y21
             Y12 Y22
             Y13 Y23;

ANALYSIS:  ESTIMATOR = ML;

MODEL:     !Messmodelle der State-Faktoren
           STATE1 BY Y11@1
             Y21@1
             ;
           STATE2 BY Y12@1
             Y22@1
             ;
           STATE3 BY Y13@1
             Y23@1
             ;
           !Varianzen der State-Faktoren
           STATE1*;
           STATE2*;
           STATE3*;

           !Parallele Messungen innerhalb der MZPe
           !Zusaetzlich zu den Faktorladungen
           !Fehlervarianzen gleichsetzen
           Y11 (e1)
           Y21 (e1)
           Y12 (e2)
           Y22 (e2)
           Y13 (e3)
           Y23 (e3);

OUTPUT:    Sampstat; STDYX;
```

Model 2: Multistate-Singletrait-Modell

```

TITLE:      Model 2: Multistate-Singletrait-Modell

DATA:      FILE = IN_Cov.txt;
           TYPE = COVARIANCE;
           NOBSERVATIONS = 302;

VARIABLE:  NAMES = Y11 Y21 Y12 Y22 Y13 Y23;

           USEVARIABLES =
             Y11 Y21
             Y12 Y22
             Y13 Y23;

ANALYSIS:  ESTIMATOR = ML;

MODEL:     !Messmodelle der State-Faktoren
           STATE1 BY Y11@1
             Y21@1
             ;
           STATE2 BY Y12@1
             Y22@1
             ;
           STATE3 BY Y13@1
             Y23@1
             ;

           !Strukturmodell
           TRAIT BY STATE1@1
             STATE2@1
             STATE3@1
             ;

           !Varianzen der State-Residuen
           STATE1*;
           STATE2*;
           STATE3*;

           !Varianz des Traitfaktors
           TRAIT*;

           ! Unkorrelierte State-Residuen untereinander
           ! und mit dem Trait
           STATE1 WITH STATE2@0 STATE3@0 TRAIT@0;
           STATE2 WITH STATE3@0 TRAIT@0;
           STATE3 WITH TRAIT@0;

           !Parallele Messungen innerhalb der MZPe
           !Zusaetzlich zu den Faktorladungen
           !Fehlervarianzen gleichsetzen
           Y11 (e1)
           Y21 (e1)
           Y12 (e2)
           Y22 (e2)
           Y13 (e3)
           Y23 (e3);

OUTPUT:    Sampstat;
           STDYX;
           MODINDICES(3.84);

```

Model 3: Multistate-Multitrait-Modell mit indikatorspezifischen Traitfaktoren

```

TITLE:      Modell 3: Multistate-Multitrait-Modell
            Mit indikatorspezifischen Traitfaktoren
            Reliabilitaetskoeffizienten

DATA:      FILE = IN_Cov.txt;
            TYPE = COVARIANCE;
            NOBSERVATIONS = 302;

VARIABLE:  NAMES = Y11 Y21 Y12 Y22 Y13 Y23;

            USEVARIABLES =
                Y11 Y21
                Y12 Y22
                Y13 Y23;

ANALYSIS:  ESTIMATOR = ML;

MODEL:     !Messmodelle der spezifischen Faktoren
            !State-Residuen
            SR1 BY Y11@1
                Y21@1
                ;
            SR2 BY Y12@1
                Y22@1
                ;
            SR3 BY Y13@1
                Y23@1
                ;

            !Messmodelle der Traitfaktoren
            TRAIT1 BY Y11@1
                Y12@1
                Y13@1;

            TRAIT2 BY Y21@1
                Y22@1
                Y23@1;

            !Varianzen der State-Residuen mit Labels
            SR1*(s1);
            SR2*(s2);
            SR3*(s3);

            !Varianzen der Traitfaktoren mit Labels
            TRAIT1*(t1);
            TRAIT2*(t2);

            TRAIT1 WITH Trait2*;

            !Korr. innerhalb der State Residuen auf null fixiert
            !Korr. zwischen SR und Traits auf null fixiert
            SR1 WITH SR2@0 SR3@0 Trait1@0 TRAIT2@0;
            SR2 WITH SR3@0 Trait1@0 TRAIT2@0;
            SR3 WITH Trait1@0 TRAIT2@0;

```

```

!Parallele Messungen innerhalb der MZPe
!Zusaetzlich zu den Faktorladungen
!Fehlervarianzen gleichsetzen
Y11 (e1)
Y21 (e1)
Y12 (e2)
Y22 (e2)
Y13 (e3)
Y23 (e3)
;

```

```

!Berechnungen der Reliabilitaetskoeffizienten

```

```

MODEL CONSTRAINT:
!Y11
!Konsistenz Y11
NEW(CON_Y11);
CON_Y11 = t1 / (t1+s1+e1);

!Spezifitaet Y11
NEW(SPE_Y11);
SPE_Y11 = s1 / (t1+s1+e1);

!Reliability Y11
NEW(REL_Y11);
REL_Y11 = (CON_Y11+SPE_Y11);

!Y21
!Konsistenz Y21
NEW(CON_Y21);
CON_Y21 = t2 / (t2+s1+e1);

!Spezifitaet Y21
NEW(SPE_Y21);
SPE_Y21 = s1 / (t2+s1+e1);

!Reliability Y21
NEW(REL_Y21);
REL_Y21 = (CON_Y21+SPE_Y21);

!Y12
!Konsistenz Y12
NEW(CON_Y12);
CON_Y12 = t1 / (t1+s2+e2);

!Spezifitaet Y12
NEW(SPE_Y12);
SPE_Y12 = s2 / (t1+s2+e2);

!Reliability Y12
NEW(REL_Y12);
REL_Y12 = (CON_Y12+SPE_Y12);

!Y22
!Konsistenz Y22
NEW(CON_Y22);
CON_Y22 = t2 / (t2+s2+e2);

!Spezifitaet Y22
NEW(SPE_Y22);
SPE_Y22 = s2 / (t2+s2+e2);

```

```
!Reliability Y22  
NEW(REL_Y22);  
REL_Y22 = (CON_Y22+SPE_Y22);
```

```
!Y13  
!Konsistenz Y13  
NEW(CON_Y13);  
CON_Y13 = t1 / (t1+s3+e3);
```

```
!Spezifitaet Y13  
NEW(SPE_Y13);  
SPE_Y13 = s3 / (t1+s3+e3);
```

```
!Reliability Y13  
NEW(REL_Y13);  
REL_Y13 = (CON_Y13+SPE_Y13);
```

```
!Y23  
!Konsistenz Y23  
NEW(CON_Y23);  
CON_Y23 = t2 / (t2+s3+e3);
```

```
!Spezifitaet Y23  
NEW(SPE_Y23);  
SPE_Y23 = s3 / (t2+s3+e3);
```

```
!Reliability Y23  
NEW(REL_Y23);  
REL_Y23 = (CON_Y23+SPE_Y23);
```

```
OUTPUT: Sampstat; STDYX;
```